

Introduction à Maxima

Richard H. Hand (Université de Cornell)

Traduction française : Michel Gosse

email : michel.gosse@ac-poitiers.fr

Version 1.1 du 3 mai 2002

Cet article est la traduction en français du document d'introduction sur Maxima, écrit par Richard Hand, de l'Université de Cornell.

Ce document a été produit par le logiciel TexMacs, dont le site est :

<http://www.texmacs.org>

qui s'interface avec Maxima et permet une présentation en notation mathématique standard des résultats produits par ce logiciel de calcul formel.

Table des matières

Table des matières	1
1 Introduction	2
2 Les touches et les symboles spéciaux	2
3 Arithmétique	4
4 Algèbre	6
5 Calcul différentiel	7
6 Calcul matriciel	9
7 Programmer avec Maxima	10
8 Une liste partielle des fonctions Maxima	10

1 Introduction

Pour lancer Maxima sous Windows, double-cliquer sur le fichier xmaxima, qui est situé dans le sous-répertoire d'installation du logiciel. Sous Linux, on peut taper dans une console soit maxima, ce qui lance le logiciel en mode texte, soit xmaxima, qui le lancera en mode graphique. Dans tous les cas, on obtient un message ressemblant à :

```
GCL (GNU Common Lisp) Version(2.4.0) Wed May 9 12:02:00 CDT 2001
Licensed under GNU Library General Public License
Contains Enhancements by W. Schelter
Maxima 5.6 Wed May 9 12:01:49 CDT 2001 (with enhancements by W. Schelter).
Licensed under the GNU Public License (see file COPYING)
```

(C1)

(C1) est une étiquette. Chaque entrée ou sortie est numérotée et on peut s'y référer par son étiquette durant toute une session maxima. Le C indique une entrée (Command), et le D une sortie (Display). Il ne faudra donc jamais employer de variables telles que C1 ou D5, car cela prêterait à confusion.

Maxima ne distingue pas les caractères en majuscule et en minuscule; Si vous entrez $\sin(x)$ ou $SIN(x)$, cela désignera la fonction sinus pour le logiciel. Par contre, pour les réponses qu'il renvoie, le logiciel emploiera systématiquement une notation en majuscule. Cette règle ne s'applique pas aux variables définies par l'utilisateur; par exemple, les variables x et X sont distinctes pour Maxima. Testez-le ...

2 Les touches et les symboles spéciaux

1. Pour arrêter une session maxima, taper `quit()`; Si vous tapez CTRL C, alors voici ce que Maxima va afficher :

```
Correctable error : Console interrupt
Signalled by Macysgma-top-level
If continued : Type :r to resume execution, or :q to quit to top level
Broken at SYSTEM:TERMINAL-INTERRUPT
Type :H for Help
Maxima> >:q
```

A l'invite de Maxima, il faut donc taper `:q` (ou `:t`) pour arrêter le calcul en cours et retourner au prompt de Maxima. Si on entre `:r`, le calcul continuera. L'appui sur les touches CTRL Y n'aura aucun effet à part celui d'être indiqué à l'écran. Enfin, CTRL Z aura le même effet que la commande `quit()`; On note CTRL C l'appui simultané sur les touches CONTROLE et C du clavier.

2. Pour arrêter un calcul sans quitter Maxima, tapez CTRL C. Il est important de connaître ce que l'on doit faire dans ce cas, qui peut se produire si vous lancez un calcul qui dure trop longtemps. Rappelez-vous de taper `:q` à l'invite `MAXIMA>>` qui apparaîtra, afin de revenir à l'écran de saisie des commandes de Maxima.
3. Afin de signaler à Maxima la fin d'une entrée de commande, il faut mettre à la fin de la saisie un point-virgule, puis appuyer sur la touche ENTREE. Remarquez que le seul appui sur la touche ENTREE n'est pas suffisant pour faire exécuter une commande par Maxima.
4. Une autre façon de signaler la fin d'une commande est d'utiliser le symbole dollar, qui supprimera l'affichage du résultat du calcul. Ceci est utile si vous évaluez un résultat intermédiaire très long, et que vous ne voulez pas perdre de temps à afficher ce résultat à l'écran.
5. Si vous voulez répéter une commande antérieure, par exemple celle désignée par l'invite (C5), vous pouvez le faire sans retaper la commande en faisant précéder l'étiquette par deux simples quotes ("), ce qui donnera "C5. Notez que si vous évaluez juste C5, cela ne correspondra pas à l'action recherchée.
6. Si vous voulez vous référer au résultat du calcul précédent fait par Maxima, vous pouvez soit utiliser son étiquette D, soit utiliser le symbole pourcentage (%).
7. Les constantes e (exponentielle de 1), i (le nombre complexe tel que $i^2 = -1$) et π sont désignées par respectivement %E (ou %e), %I (ou %i) et %PI (ou %pi). Remarquez que le % employé ici n'a aucun rapport avec le % désignant le résultat précédemment calculé par Maxima.
8. Afin d'affecter une valeur à une variable, Maxima utilise le signe deux-points (:), et non pas le signe égal. Le signe égal sert à écrire une équation.

3 Arithmétique

Les opérations fondamentales sont :

- l'addition notée +

- la soustraction notée -
- la multiplication des réels notée *
- la division notée /
- l'élevation à la puissance notée ^ ou **
- la multiplication matricielle notée .
- la fonction racine carrée notée sqrt

Les sorties de Maxima se font en valeur exacte. Par exemple :

```
maxima] 1/100+1/101;
```

(D2) $\frac{201}{10100}$

Si on emploie des nombres irrationnels, alors ceux-ci sont conservés sous forme symbolique :

```
maxima] (1+sqrt(2))^5;
```

(D3) $(\sqrt{2} + 1)^5$

```
(C4) expand(%);
```

(D4) $29\sqrt{2} + 41$

Toutefois, il est souvent utile d'obtenir une valeur approchée d'un résultat. Cette opération s'effectue en faisant suivre l'expression demandée de la commande ,numer :

```
maxima] %,numer;
```

(D5) 82.01219330881976

Notez que le % se réfère ici au résultat du calcul précédent. Par défaut, la commande numer donne 16 chiffres significatifs, le dernier pouvant être erroné. Toutefois, Maxima peut calculer avec une précision arbitraire aussi importante que l'on veut grâce à la fonction bfloat :

```
maxima] bfloat(D3);
```

8.201219330881976B1

Le nombre de décimales est contrôlé par la variable Maxima FPPREC, qui a par défaut la valeur 16 :

```
maxima] fpprec;
```

(D7) 16

Assignons à cette variable la valeur 100 :

```
maxima] fpprec:100;
```

(D8) 100

```
(C9) bfloat(D3);
```

8.2012193308819756415248973002081244278520484385931494122123712401731
2418754011041266612384955016056B1

Maxima peut calculer les valeurs exactes de très grand nombres :

```
maxima] 100!;
```

```
(D12) 93326215443944152681699238856266700490715968264381621468592963895217599993229\
915608941463976156518286253697920827223758251185210916864000000000000000000000
```

4 Algèbre

L'importance de Maxima comme logiciel de calcul formel apparaît clairement quand nous voyons avec quelle facilité le logiciel va effectuer des calculs algébriques. Voici un exemple de développement d'un polynôme :

```
(C1) (x+3*y+x^2*y)^3;
```

```
(D1) (x^2 y + 3 y + x)^3
```

```
(C2) expand(%);
```

```
(D2) x^6 y^3 + 9 x^4 y^3 + 27 x^2 y^3 + 27 y^3 + 3 x^5 y^2 + 18 x^3 y^2 + 27 x y^2 + 3 x^4 y + 9 x^2 y + x^3
```

Supposons maintenant que nous voulons remplacer x par la valeur $\frac{5}{z}$ dans l'expression précédente :

```
maxima] d2,x=5/z;
```

```
(D4) \frac{135 y^2}{z} + \frac{675 y^3}{z^2} + \frac{225 y}{z^2} + \frac{2250 y^2}{z^3} + \frac{125}{z^3} + \frac{5625 y^3}{z^4} + \frac{1875 y}{z^4} + \frac{9375 y^2}{z^5} + \frac{15625 y^3}{z^6} + 27 y^3
```

La fonction RATSIMP de Maxima réduit l'expression au même dénominateur :

```
maxima] ratsimp(%);
```

```
(D5) \frac{27 y^3 z^6 + 135 y^2 z^5 + (675 y^3 + 225 y) z^4 + (2250 y^2 + 125) z^3 + (5625 y^3 + 1875 y) z^2 + 9375 y^2 z + 15625 y^3}{z^6}
```

Les expressions peuvent être factorisées :

```
maxima] factor(%);
```

```
(D6) \frac{(3 y z^2 + 5 z + 25 y)^3}{z^6}
```

Maxima peut trouver les valeurs exactes des solutions de systèmes d'équations algébriques, même dans des cas non linéaires. Dans l'exemple suivant, on résout trois équations dont les inconnues sont a , b et c :

```
maxima] a+b*c=1;
```

```
(D7) b c + a = 1
```

```
(C8) b-a*c=0;
```

$$(D8) \quad b - a c = 0$$

$$(C9) \quad a + b = 5;$$

$$(D9) \quad b + a = 5$$

$$(C10) \quad \text{solve}([d7, d8, d9], [a, b, c]);$$

$$(D10) \quad \left[\left[a = \frac{25\sqrt{79}i + 25}{6\sqrt{79}i - 34}, b = \frac{5\sqrt{79}i + 5}{\sqrt{79}i + 11}, c = \frac{\sqrt{79}i + 1}{10} \right], \left[a = \frac{25\sqrt{79}i - 25}{6\sqrt{79}i + 34}, b = \frac{5\sqrt{79}i - 5}{\sqrt{79}i - 11}, c = -\frac{\sqrt{79}i - 1}{10} \right] \right]$$

Remarquez que la sortie consiste en une " liste ", c'est à dire en une expression comprise entre deux crochets [...], qui contient elle-même deux listes. Chacune de ces dernières contient une solution au système d'équations.

Les formules trigonométriques se manipulent aisément avec Maxima. La fonction TRIGEXPAND utilise la formule de la somme afin de simplifier au maximum l'argument de chacune des fonctions trigonométriques utilisées :

$$(C1) \quad \sin(u+v) * \cos(u)^3;$$

$$(D1) \quad \cos^3 u \sin(v + u)$$

$$(C2) \quad \text{trigexpand}(%);$$

$$(D2) \quad \cos^3 u (\cos u \sin v + \sin u \cos v)$$

Inversement, la fonction TRIGREDUCE transforme une expression trigonométrique en une somme de termes dont chacun ne comporte qu'une seule fois la fonction sinus ou la fonction cosinus :

$$\text{maxima] } \text{trigreduce}(d1);$$

$$(D3) \quad \frac{\sin(v + 4u) + \sin(v - 2u)}{8} + \frac{3\sin(v + 2u) + 3\sin v}{8}$$

Les fonctions REALPART et IMAGPART renvoient respectivement la partie réelle et la partie imaginaire d'un nombre complexe :

$$\text{maxima] } w:3+k*i;$$

$$(D4) \quad i k + 3$$

$$(C5) \quad w^2 * e^w;$$

$$(D5) \quad (i k + 3)^2 e^{ik+3}$$

$$(C6) \quad \text{realpart}(%);$$

$$(D6) \quad e^3 (9 - k^2) \cos k - 6 e^3 k \sin k$$

$$(C7) \quad \text{imagpart}(D5);$$

$$(D7) \quad e^3 (9 - k^2) \sin k + 6 e^3 k \cos k$$

5 Calcul différentiel

Maxima calcule dérivées et intégrales, effectue des développements en série de Taylor, trouve les limites, et sait résoudre en valeur exacte les équations différentielles ordinaires. Nous commençons par définir la fonction f de la variable numérique x par :

```
maxima] f:x^3*e^(k*x)*sin(a*x);
```

$$(D9) \quad x^3 e^{kx} \sin(ax)$$

On calcule l'expression de la dérivée de f par rapport à x :

```
maxima] diff(f,x);
```

$$(D10) \quad kx^3 e^{kx} \sin(ax) + 3x^2 e^{kx} \sin(ax) + ax^3 e^{kx} \cos(ax)$$

Maintenant, on calcule une primitive de f par rapport à x :

```
maxima] integrate(f,x);
```

$$(D11) \quad \frac{(k^7 + 3a^2k^5 + 3a^4k^3 + a^6k)x^3 + (-3k^6 - 3a^2k^4 + 3a^4k^2 + 3a^6)x^2 + (6k^5 - 12a^2k^3 - 18a^4k)x - 6k^4 + 36a^2k^2 - 6a^4}{k^8 + 4a^2k^6 + 6a^4k^4 + 4a^6k^2} e^{kx} \sin(ax) + \left(\frac{-ak^6}{k^8 + 4a^2k^6 + 6a^4k^4 + 4a^6k^2} \right)$$

Un léger changement de syntaxe fournit l'intégrale définie :

```
maxima] integrate(1/x^2,x,1,inf);
```

$$(D12) \quad 1$$

```
(C13) integrate(1/x,x,0,inf);
```

Integral is divergent

```
-- an error. Quitting. To debug this try DEBUGMODE(TRUE);)
```

Ensuite, nous définissons la fonction g à l'aide de la fonction f et du sinus hyperbolique, puis nous trouvons le développement en série de Taylor de g à l'ordre 3 au voisinage de $x=0$.

```
maxima] g:f/sinh(k*x)^4;
```

$$(D14) \quad \frac{x^3 e^{kx} \sin(ax)}{\sinh^4(kx)}$$

```
(C15) taylor(g,x,0,3);
```

$$(D15) \quad \frac{a}{k^4} + \frac{ax}{k^3} - \frac{(ak^2 + a^3)x^2}{6k^4} - \frac{(3ak^2 + a^3)x^3}{6k^3} + \dots$$

La limite de g quand x tend vers 0 se calcule par l'instruction suivante :

```
maxima] limit(g,x,0);
```

$$(D16) \quad \frac{a}{k^4}$$

Maxima peut aussi représenter des dérivées sous une forme non évaluée. On remarquera le signe quote ('') devant l'expression :

```
maxima] 'diff(y,x);
```

$$(D17) \quad \frac{d}{dx} y$$

L'opérateur quote (') dans l'expression précédente signifie " ne pas évaluer ". Sans ce symbole, Maxima aurait retourné 0 :

```
maxima] diff(y,x);
```

(D18) 0

En utilisant ce symbole quote ('), on peut écrire des équations différentielles :

```
maxima] 'diff(y,x,2)+'diff(y,x)+y;
```

(D19) $\frac{d^2}{dx^2}y + \frac{d}{dx}y + y$

La fonction ODE2 de Maxima peut résoudre des équations différentielles du premier et second ordre :

```
maxima] ode2(D19,y,x);
```

(D20) $y = e^{-\frac{x}{2}} \left(\%K1 \sin\left(\frac{\sqrt{3}x}{2}\right) + \%K2 \cos\left(\frac{\sqrt{3}x}{2}\right) \right)$

%K1 et %K2 désignent des constantes réelles.

6 Calcul matriciel

Maxima peut calculer le déterminant, l'inverse, les valeurs propres et les vecteurs propres d'une matrice qui comporte des éléments symboliques (c'est à dire des expressions avec des variables algébriques). On commence par entrer une matrice m élément par élément :

```
maxima] m:entermatrix(3,3);
```

Is the matrix 1. Diagonal 2. Symmetric 3. Antisymmetric 4. General

```
Answer 1, 2, 3 or 4 : 4;
```

```
Row 1 Column 1: 0;
```

```
Row 1 Column 2: 1;
```

```
Row 1 Column 3: a;
```

```
Row 2 Column 1: 1;
```

```
Row 2 Column 2: 0;
```

```
Row 2 Column 3: 1;
```

```
Row 3 Column 1: 1;
```

```
Row 3 Column 2: 1;
```

```
Row 3 Column 3: 0;
```

```
Matrix entered.
```

(D21) $\begin{pmatrix} 0 & 1 & a \\ 1 & 0 & 1 \\ 1 & 1 & 0 \end{pmatrix}$

On calcule ensuite sa transposée, son déterminant et son inverse :

```
maxima] transpose(m);
```

(D22) $\begin{pmatrix} 0 & 1 & 1 \\ 1 & 0 & 1 \\ a & 1 & 0 \end{pmatrix}$

(C23) `determinant(m);`

(D23) $a + 1$

(C24) `invert(m),detout;`

$$(D24) \frac{\begin{pmatrix} -1 & a & 1 \\ 1 & -a & a \\ 1 & 1 & -1 \end{pmatrix}}{a + 1}$$

Dans l'entrée C24, l'option `detout` impose la factorisation par le déterminant, ce dernier restant à l'extérieur de la matrice. Comme vérification, on peut multiplier m par son inverse (remarquer l'usage du point $(.)$ pour désigner le produit matriciel) :

`maxima] m.d24;`

$$(D25) \begin{pmatrix} 0 & 1 & a \\ 1 & 0 & 1 \\ 1 & 1 & 0 \end{pmatrix} \cdot \frac{\begin{pmatrix} -1 & a & 1 \\ 1 & -a & a \\ 1 & 1 & -1 \end{pmatrix}}{a + 1}$$

(C26) `expand(%);`

$$(D26) \begin{pmatrix} \frac{a}{a+1} + \frac{1}{a+1} & 0 & 0 \\ 0 & \frac{a}{a+1} + \frac{1}{a+1} & 0 \\ 0 & 0 & \frac{a}{a+1} + \frac{1}{a+1} \end{pmatrix}$$

(C27) `factor(%);`

$$(D27) \begin{pmatrix} 1 & 0 & 0 \\ 0 & 1 & 0 \\ 0 & 0 & 1 \end{pmatrix}$$

Pour trouver valeurs et vecteurs propres, on utilise la commande `EIGENVECTORS` :

`maxima] eigenvectors(m);`

Warning - you are redefining the MACSYMA function EIGENVALUES

Warning - you are redefining the MACSYMA function EIGENVECTORS

$$(D28) \left[\left[\left[-\frac{\sqrt{4a+5}-1}{2}, \frac{\sqrt{4a+5}+1}{2}, -1 \right], [1, 1, 1] \right], \left[1, -\frac{\sqrt{4a+5}-1}{2a+2}, -\frac{\sqrt{4a+5}-1}{2a+2} \right], \left[1, \frac{\sqrt{4a+5}+1}{2a+2}, \frac{\sqrt{4a+5}+1}{2a+2} \right], [1, -1, 0] \right]$$

Dans l'expression D28, le premier crochet à l'intérieur du triple crochet donne les trois valeurs propres, le suivant leur ordre de multiplicité respectifs, et les crochets suivants fournissent les coordonnées des vecteurs propres associés aux valeurs propres. Afin d'extraire un vecteur propre de l'expression précédente, on utilise la commande `PART` :

`maxima] part(D28,2);`

$$(D29) \left[1, -\frac{\sqrt{4a+5}-1}{2a+2}, -\frac{\sqrt{4a+5}-1}{2a+2} \right]$$

7 Programmer avec Maxima

Jusqu'à présent, nous avons utilisé Maxima en mode interactif, comme une super calculatrice. Toutefois, pour des calculs qui nécessitent une séquence de commandes répétitive, il est préférable d'exécuter un programme. Nous présentons maintenant un exemple de court programme permettant le calcul des points critiques d'une fonction f de deux variables x et y . Le programme demande à l'utilisateur d'entrer une fonction f , puis il calcule les dérivées partielles $\frac{\partial f}{\partial x}$ et $\frac{\partial f}{\partial y}$, et enfin il fait appel à la commande SOLVE de Maxima pour obtenir les solutions du système d'équations :

$$\frac{\partial f}{\partial x} = 0 \text{ et } \frac{\partial f}{\partial y} = 0$$

Le programme est écrit avec un éditeur de textes quelconque, puis est chargé dans Maxima avec la commande BATCH. Voici le listing du programme :

```

/*-----*/
Ceci est le programme critpts.max
comme vous le voyez, les commentaires avec Maxima sont analogues à ceux en langage C
Auteur : Nelson Luis Dias
Créé le 07/07/2000
-----*/

critpts():=(
print("Programme pour déterminer les points critiques "),
/*-----*/
On entre une fonction
-----*/
f:read("Entrez f(x,y)"),
/*-----*/
On renvoie la réponse pour vérification
-----*/
print("f = ",f),
/*-----*/
On construit une liste avec les dérivées partielles :
-----*/
eqs:[diff(f,x),diff(f,y)],
/*-----*/
On construit la liste des inconnues
-----*/
unk:[x,y],
/*-----*/
On résout le système
-----*/
solve(eqs,unk)
)
$

```

Le programme (qui est en l'état actuel une fonction sans argument) s'appelle critpts. Chaque ligne est une commande (qui doit être valide) Maxima qui pourrait être exécutée en mode interactif, et qui est séparée de la commande suivante par une virgule. Les dérivées partielles sont stockées dans une liste qui s'appelle eqs, et les inconnues dans une liste qui s'appelle unk. Voici un exemple d'exécution du programme :

```

(C1) batch("/home/michel/temp/critpts.max");
batching #p/home/michel/temp/critpts.max
(C2) critpts() := (PRINT("Programme recherchant les points critiques "),
f : READ("Entrez f(x,y)"), PRINT(" f = ", f), eqs : [DIFF(f, x), DIFF(f, y)],

```

unk : [x, y], SOLVE(eqs, unk))

(C3) critpts();

Programme recherchant les points critiques

Entrez f(x,y)

%E^(x^3+y^2)*(x+y);

$f = (y + x)e^{y^2+x^3}$

(D3) [[x = 0.4588955685487 %I + 0.35897908710869,

y = 0.49420173682751 %I - 0.12257873677837],

[x = 0.35897908710869 - 0.4588955685487 %I,

y = - 0.49420173682751 %I - 0.12257873677837],

[x = 0.41875423272348 %I - 0.69231242044203,

y = 0.86972626928141 %I + 0.4559120701117]]

y = 0.4559120701117 - 0.86972626928141 %I],

[x = - 0.41875423272348 %I - 0.69231242044203,

y = 0.86972626928141 %I + 0.4559120701117]]

8 Une liste partielle des fonctions Maxima

Vous pouvez consulter le manuel Maxima dans le sous-répertoire maxima-/info au format texinfo ou html. A l'intérieur de Maxima, vous pouvez utiliser la commande DESCRIBE(nom de la fonction).

- ALLROOTS(A)

Trouve toutes les solutions en valeur approchée (avec 16 chiffres significatifs par défaut), éventuellement dans le corps des nombres complexes, d'une équation polynomiale A, et en donne une liste.
- APPEND(A,B)

Réunit la liste A et la liste B, et donc renvoie une liste unique.
- BATCH(A)

Charge et lance un programme BATCH du nom de A
- COEFF(A,B,C)

Renvoie le coefficient de B puissance C dans l'expression A
- CONCAT(A,B)

Crée le symbole AB
- CONS(A,B)

Ajoute A à la liste B en première position
- DEMOIVRE(A)

Transforme toutes les exponentielles complexes sous forme trigonométrique
- DENOM(A)

Renvoie le dénominateur de A
- DEPENDS(A,B)

Cette commande indique que A est une fonction de B. Cette dernière est utile pour écrire des dérivées de manière générale, comme par exemple dans les équations différentielles.
- DESOLVE(A,B)

Essaie de résoudre un système linéaire A d'équations différentielles d'inconnues B en utilisant la transformée de Laplace
- DETERMINANT(A)

Renvoie le déterminant de la matrice carrée A
- DIFF(A,B1,C1,B2,C2,...,Bn,Cn)

Renvoie les dérivées partielles de A par rapport aux variables B1 à la puissance C1, B2 à la puissance C2, etc..; Un raccourci pour DIFF(A,B,1) est DIFF(A,B). 'DIFF(A,B) représente la dérivée sous forme non évaluée, ce qui est utile pour les équations différentielles.

- EIGENVALUES(A)

Renvoie deux listes, la première étant les valeurs propres d'une matrice carrée A, et la seconde leur ordre respectif de multiplicité.
- EIGENVECTORS(A)

Fait la même chose que la commande précédente, mais rajoute la liste des vecteurs propres associés
- ENTERMATRIX(A,B)

Demande à l'utilisateur d'entrer une matrice n sur m, élément par élément
- EV(A,B1,B2,...,Bn)

Évalue A en tenant compte des conditions B1, B2, ... Bn. En particulier, les Bi peuvent être des équations, des listes d'équations (comme ce que retourne la commande SOLVE), ou bien des assignations, c'est à dire que dans ce cas la commande EV remplace les Bi par leurs valeurs dans A. Les Bi peuvent être des mots comme NUMER (ce qui dans ce cas renvoie une valeur numérique pour A), DETOUT (ce qui dans ce cas entraîne que tout calcul de matrice inverse se fera en gardant le déterminant en facteur, c'est à dire à l'extérieur de la matrice), ou DIFF (et dans ce cas tous les calculs de dérivation sont effectués, c'est à dire que 'DIFF est remplacé par DIFF). Pour abrégé une commande interactive, c'est à dire qui n'est pas incluse dans la définition d'une fonction, la commande EV peut être omise, ce qui raccourcit la commande sous la forme : A,B1,B2,...,Bn
- EXPAND(A)

Effectue le développement de A. En particulier, les multiplications sont distribuées par rapport aux additions.
- EXPONENTIALIZE(A)

Transforme toutes les fonctions trigonométriques contenues dans A en exponentielles complexes.
- FACTOR(A)

Factorise l'expression A
- FREEOF(A,B)

Renvoie TRUE si l'expression B ne comporte pas la variable A
- GRIND(A)

Ecrit une variable ou une fonction A sous une forme compacte. Si cette commande est utilisée avec WRITEFILE et un éditeur de textes, cela permet de produire des fichiers BATCH incluant des expressions générées par Maxima.
- IDENT(A)

Renvoie la matrice identité d'ordre A
- IMAGPART(A)

Renvoie la partie imaginaire de A
- INTEGRATE(A,B)

Essaie de déterminer une primitive de A par rapport à B
- INTEGRATE(A,B,C,D)

Essaie de calculer l'intégrale de A par rapport à B entre les bornes C et D. Les bornes de l'intégrale peuvent être INF (plus l'infini) ou MINF (moins l'infini).
- INVERT(A)

Calcule la matrice inverse de la matrice carrée A
- KILL(A)

Élimine la variable A de toutes les définitions et propriétés de l'environnement actuel de Maxima

- **LIMIT(A,B,C)**
Donne la limite de l'expression A quand B tend vers C. Cette dernière variable peut prendre comme valeur INF (plus l'infini) ou MINF (moins l'infini), comme pour l'intégrale définie.
- **LHS(A)**
Donne le membre de gauche d'une équation A
- **LOADFILE(A)**
Charge un fichier du nom de A situé dans le répertoire courant. Le fichier doit être dans un format adapté, c'est à dire avoir été auparavant créé par la commande SAVE
- **MAKELIST(A,B,C,D)**
Crée une liste d'éléments A (chacun d'eux dépendant probablement de B) pour B variant de C à D
- **MAP(A,B)**
Applique la fonction A à la sous-expression B
- **MATRIX(A1,A2,...An)**
Crée la matrice dont les lignes sont les Ai, chacun des Ai étant une liste de m éléments [B1,B2,...Bm]
- **NUM(A)**
Renvoie le numérateur de A
- **ODE2(A,B,C)**
Essaie de résoudre l'équation différentielle du premier ou du second ordre A d'inconnue la fonction B dont la variable est C
- **PART(A,B1,B2,...Bn)**
Prend la B1 ième partie de A, puis la B2 ième partie de A, et ainsi de suite
- **PLAYBACK(A)**
Affiche les dernières A (entiers) sorties produites par Maxima. Si A n'est pas fourni, alors toutes les lignes depuis le début de la session sont affichées
- **RATSIMP(A)**
Simplifie A et renvoie le quotient de deux polynômes
- **REALPART(A)**
Renvoie la partie réelle de A
- **RHS(A)**
Donne le membre de droite d'une équation A
- **SAVE(A,B1,B2,...Bn)**
Crée un fichier dont le nom est A dans le sous-répertoire par défaut, contenant les variables, les fonctions ou les tableaux Bi. Le format de ce fichier permet son chargement ultérieur dans Maxima en utilisant la commande LOADFILE. Toute la session, incluant les étiquettes, peut être sauvegardée en choisissant B1 égal à ALL.
- **SOLVE(A,B)**
Essaie de résoudre l'équation algébrique A d'inconnue B. Une liste des solutions est renvoyée. Par simplification, si A est une équation de la forme C=0, alors il est possible de n'écrire que C seulement.
- **STRING(A)**
Convertit A en notation linéaire Maxima, analogue à celle utilisée en Fortran, comme si l'expression avait été entrée directement, et mets A dans un buffer pour une édition possible. L'expression ainsi obtenue ne doit pas être employée pour des calculs ultérieurs.
- **STRINGOUT(A,B1,B2,...,Bn)**

Crée un fichier dont le nom est A dans le sous-répertoire courant, contenant les variables B1, B2,..Bn. Ce fichier est au format texte et il est possible de le charger avec la commande BATCH sous Maxima. On peut de cette manière sauvegarder une session Maxima. De plus, les expressions sauvegardées ainsi peuvent être réutilisées dans un programme en Fortran, en Basic ou en C avec quelques modifications.

- SUBST(A,B,C)
Remplace la valeur A par B dans C
- TAYLOR(A,B,C,D)
Développe A en série de Taylor par rapport à la variable B au voisinage de C, jusqu'à l'ordre D, en incluant le terme $(B-C)^D$. Maxima peut aussi effectuer des développements en série de Taylor avec plus d'une variable indépendante. Consulter le manuel Maxima.
- TRANSPOSE(A)
Renvoie la transposée de la matrice A
- TRIGEXPAND(A)
Est une fonction de simplification trigonométrique qui utilise la formule de la somme des angles pour simplifier les arguments des fonctions SIN et COS rencontrées. Par exemple, TRIGEXPAND(SIN(x+y)) donne $\text{COS}(x)\text{SIN}(y)+\text{SIN}(x)\text{COS}(y)$.
- TRIGREDUCE(A)
Est une fonction de simplification trigonométrique qui utilise des égalités trigonométriques pour convertir des produits et des puissances de SIN et COS en une somme de termes, chacun d'eux ne contenant qu'une seule fois la fonction SIN ou COS. Par exemple, TRIGREDUCE(SIN(x)^2) donne $(1-\text{COS}(2x))/2$.
- TRIGSIMP(A)
Est une fonction de simplification trigonométrique qui remplace TAN, SEC, etc, par leurs équivalents en SIN et COS. L'égalité $\text{SIN}(x)^2 + \text{COS}(x)^2 = 1$ est aussi utilisée.